EMU41 Update 2006
Again Jean-Francois Garnier helpfully adds new features to his nice EMU41 emulator software : Up to now, for advanced applications like handling rom-images, creating own user code modules or machine code development, EMU41 emulated a 8KByte MLDL RAM page and the famous W&W RAM-Box. This virtual RAM-Box consists of two software selected 32 KByte banks = 64KByte RAM. 
ERAMCO 128KByte RSU

Now Jean-Francois also emulates the Eramco RSU RAM-Storage-Unit which supports four software selected 32 KByte banks = 128KByte RAM. The operating system for the W&W RAM-Box is stored in a 4KByte block. Similarly the Eramco operation system is stored in a 8KByte block and includes additionally commands for data storing applications and its own text editor. In my experience the 128KByte RSU from Eramco is the greatest "classic" RAM-Box solution, designed for the HP41 system. I have also heard rumours about an extended 256 KByte RSU solution (8 software selected banks) but have never seen this. To activate the Eramco 128Kbyte RSU to EMU41, edit the EMU41.INI file and add the following extensions :

  8  96  es_rsu.dat    ; the 96k RAM = 4 banks of 24k in pages 8 to 13

14  8    rsu_os.rom  ; the RSU OS (supporting the multiple banks)

In this case the 8 KByte RSU operating system is addressed to port 14 and port 15. If so, as Jean-Francois explained, it would be possible to override RSU pages by plug in modules. In the following example we place the CCD-Module to port 8 and port 9 

  8  96  es_rsu.dat

14  8    rsu_os.rom

  8  8    ccd1b.bin        ; override pages 8 & 9

When operating the RSU, remember that Eramco numbers the plug in port pages in an own way : W&W counts from page 8 up to page 15, identical to the Zenrom page numbering. In contrast to this, Eramco start with +1 for page 8, -1 for page 9 and ends with +4 for page 14 and -4 for page 15.

RS232 support

Beside this great implementation of the Eramco 128KByte RSU (which works with its own bank switching mechanism) Jean-Francois also adds a virtual 
HP-IL/RS232 interface device to EMU41, supporting the serial COM port of the PC. Now EMU41 includes the following list of virtual HP-IL devices (taken from the EMU41.INI file) :

[DEVICES]
;

DISPLAY

LPRTER1

HDRIVE1

;HDRIVE2

SERIAL1 1,E3

;SERIAL2 2,E3

;FDRIVE1       ;  insert a LIF DD floppy for correct function. Not available for modern PCs.

DOSLINK
XIL 300           ;   HP-IL/PC Interface Card - 700 default address / 300 address for WIN2000/NT/XP

The new SERIAL1 device acts like a virtual HP82164A HP-IL/RS232 interface for EMU41 and works as a gateway to you real HP41 handheld. Use this to transfer the X-Register contents, programs, data register blocks, data-files and matrix-files (CCD-Module) between EMU41 and your HP41 handheld computer. For your real HP41 you need the IL-Module, EXT-I/O and CCD-Module plus the HP-IL/RS232 interface HP82164A or CMT-RAM-DISC, wired to the serial COM port of the PC :

This new RS232 gateway solution between HP41 and EMU41 is handy for PC configurations that can not use the HP-IL/PC Interface Card : For example notebook- and laptop-PC´s and PC´s with motherboards without  an ISA-slot. As Jean-Francois explains, it would be also possible to add a RS232-to-USB Bus interface : With this expanded configuration you get a gateway solution for PC hardware that only supports the USB-Bus.

In contrast to the "classic" HP-IL/PC Interface Card gateway solution, the new RS232 application is limited to bidirectional program- and data transfer. Using the new RS232 gateway you are not able to control real HP-IL devices by EMU41. Controlling real HP-IL devices like the IL-Digitalvoltmeter or the IL-Plotter by EMU41 is only possible by using the HP-IL/PC Interface Card. 

HP41 software

In order to use the new RS232 gateway for interfacing EMU41 to your real HP41
I created a software pack plus associated user documentation. You can find this on the HPCC website. Because HP41 and EMU41 are running at different speeds (depending on the PC), and no COM data buffer exist on PC side for holding the transferred data for EMU41, we have a time critical application. Therefore the correct system configuration and time alignment is needed on the PC side, as described below..

For data transfer we execute the EXT-I/O Module commands OUTAN and INAN because only from this results a correct and save data transfer for this RS232 application. This two commands transfer a dummy-byte D followed by 7 data-bytes from the Alpha-Register of the sending device to the Alpha-Register of the receiving device. To exchange the 7 data-bytes between Alpha-Register and X-Register we use the (synthetic) commands STO M and RCL M. With CCD-Module you enter this commands as STO ∙ M and RCL ∙ M. For placing the leading dummy-byte D to the Alpha-Register we use the string "D6543210". 

For program download from EMU41 to HP41 we use the standard commands OUTP and INP. Because of the critical timing needed for program upload from HP41 to EMU41 we can not execute the commands OUTP and INP. Instead we use the advanced commands PEEKB and POKEB from CCD-Module for the transfer of program bytes to the X-Register. Next STO M and RCL M transfer the program bytes to the Alpha-Register, and the commands OUTAN and INAN execute the RS232 transfer. Generally start the program upload from HP41 to EMU41 with GTO ∙ ∙ on the sending and receiving side. The EMU41 software routine RECVP for receiving program code acts as a self programming program. After the program code transfer is complete, RECVP writes the new .END. position to status register c. When RECVP is finished – again you need a manual PACK = GTO ∙ ∙. Beside the data or program values in the X-Register, the programs below needs some control numbers for controlling program loops or for calculating absolute addresses. All this numbers are managed completely by the stack lift – therefore no data register are needed. 

PC COM Port configuration

On PC side open the device manager for changing the settings needed for the serial COM 1 port :
9600 Byte per second





8 data bytes





no parity
1 stop bit

protocol type = None or XON/XOFF     depends on PC

extended settings :

using FIFO buffer                        depends on PC

buffer size = low … high             depends on PC

The first four settings you select with the device manager are fixed.. The needed protocol type and the definition of possible extended settings depending from your PC system. To reach the correct transfer of data please execute the short program tests in the next chapters :
EMU41 configuration
Open the EMU41.INI file and add the following extensions for activating the SERIAL1 or SERIAL2 feature. Furthermore you have to implement the IL-Module, the EXT-I/O and CCD-Module :

[MODULES]

;

; HP-41CX Rom-Modules :

0 12 rom41cx.bin

3  4 cxfcn2d.bin

5  8 cxtime2c.bin

;

; HP-IL Module : hard addressed module to port 6 + 7

6  4 ilprin2e.bin    ;   Printer              system rom XROM 29

7  4 ilmass1h.bin  ;   Mass Storage   system rom XROM 28

;

; HP-IL Accessory Modules :

10 4 extio1a.bin   ;      EXT-I/O        XROM 23

;

; Advanced-Modules :

8 8 ccd1b.bin      ;        CCD             XROM 09 + XROM 11

;

 [DEVICES]

;

LPRTER1

HDRIVE1

HDRIVE2

SERIAL1 1,E3

;SERIAL2 2,E3

;FDRIVE1          ;   insert a LIF DD floppy for correct function. Not available for modern PCs.

DOSLINK
XIL 300           ;   HP-IL/PC Interface Card - 700 default address,

;                                                                       300 address for WIN2000/NT/XP

;
Software Commands for program- and data-transfer
The following software commands are available for program and data transfer between HP41 and EMU41 using the RS232 gateway :



SENDX
Send X-Register 


SENDP
Send user code program


SENDR
Send main memory register block



SENDF
Send X-Memory data-file



SENDM
Send CCD-Module matrix-file



RECVX
Receive X-Register


RECVP
Receive user code program



RECVR
Receive main memory register block



RECVF
Receive X-Memory data-file



RECVM
Receive CCD-Module matrix-file

Starting with a first transfer test

For starting a first transfer test enter the program codes for SENDX and RECVX including the belonging subroutines LBL 20 (initialisation) and LBL 30 (waiting routine) and LBL 41 (end routine) to HP41 and to EMU41. Note that the program codes for SENDP and for RECVP, for LBL 20 and LBL 30 are different for HP41 and for EMU41. 

Now start a X-Register upload from HP41 to EMU41 : Execute RECVX on EMU41. Next enter a number to the HP41 X-Register and execute SENDX on the HP41. In the best case now you find this number also in the EMU41 screen. If not change some settings on PC side : May be LBL 30 runs in an endless loop, than your PC is not able to use this subroutine – in this case remove it. Next change the settings needed for the COM port in the device manager as described above. Last you are able to speed up EMU41 by pressing the tab-key. This would be helpfully for older PC systems. If X-Register data upload from HP41 to EMU41 works correctly go on with a X-Register download from EMU41 to HP41. 

Register-Block transfer

In the next step you execute a register-block upload from HP41 main memory to EMU41 main memory. Enter a control number bbb.eee on both sides, bbb is the first register and eee is the last register you want to transfer. Begin with an upload of only 10 data-registers from HP41 to EMU41 : First execute RECVR on EMU41 than SENDR on the HP41. Generally the data register upload from HP41 to EMU41 needs no time alignment because the PC waits for data coming from HP82164A. 

If the results on EMU41 side are correct, execute a register-block download from EMU41 to HP41. Depending on your PC system you have to modify LBL 50 which works as a software delay routine : When your PC runs too fast, you have to insert a larger number of PSE commands. This slows down the sending routines of EMU41. You reach the correct time alignment when both routines SENDR and RECVR on both sides are finished at nearly the same time. Check this time alignment also by downloading a greater number of data registers from EMU41 to HP41. The following routines are working with the identical time alignment.
Data-File transfer

For data-file transfer create the needed X-Memory data-file or CCD-Module matrix-file with correct size on the receiving HP41 or EMU41. Next enter the file name to Alpha-Register on both HP41 and EMU41 Now start the receive routine RECVF or RECVM on the receiving device. Last start the send routine SENDF or SENDM on the sending device.
Program transfer

Program download form EMU41 to HP41 is easy, first start RECVP on the HP41 than execute SENDP on EMU41 and enter the program name at the Alpha prompt followed by R/S.

If you want to upload a program from HP41 to EMU41, first check if enough storage register exist to accommodate the transferred program on the receiving EMU41 side by entering the PRGM mode,. If not enough register exist change SIZE or delete an other program from main memory. When you make not enough memory space available, than EMU41 runs to a MEMORY LOST. Next execute GTO ∙ ∙ on both sides. Than start RECVP on EMU41. Now start SENDP on HP41, enter the program name and press R/S. When the transfer is finished you have to execute a manual PACK = GTO ∙ ∙ on EMU41 !

Software for HP41 and HP82164A
 01_LBL "RS232"
 02_LBL "SENDX


send X-Register

 03 XEQ 20



initialisation of the HP82164A
 04 "D6543210"


loaf dummy Bytes to Alpha
 05 STO M



STO ∙ M replaces dummy Bytes with 
 06 7




X-Register contents
07 OUTAN



send Alpha-Register
 08 RDN

 09 GTO 41



end routine
 10 RTN

 11_LBL "SENDP"


first execute GTO ∙ ∙ than send program
 12 XEQ 20



initialisation of the HP82164A
 13 "PROG NAME ?"

asks the program name
 14 AON

 15 PROMPT

 16 AOFF

 17 PHD



absolute address of the first program Byte
 18 ENTER^

 19 ENTER^

 20 PPLNG

 21 2

 22 -

 23 "D6543210"

 24 STO M

 25 7

 26 OUTAN



send program length in Bytes
 27 RDN

 28 CHS

 29 A+B

 30 X<>Y

 31_LBL 03

 32 PEEKB

 33 STO M



STO ∙ M
 34 7

 35 OUTAN



send program Byte
 36 RDN

 37 RDN

 38 A-

 39 X≠Y?

 40 GTO 03

 41 GTO 40

 42 RTN

 43_LBL "SENDR"


send data register block
 44 XEQ 20

 45 "D6543210"

 46_LBL 00

 47 RCL IND X

 48 STO M



STO ∙ M
 49 7

 50 OUTAN

 51 RDN

 52 RDN

 53 ISG X

 54 GTO 00

 55 GTO 40

 56 RTN

 57_LBL "SENDF"


send X-Memory data-file
 58 CLX

 59 SEEKPTA

 60 XEQ 20

 61 "D6543210"

 62_LBL 01

 63 SF 25

 64 GETX

 65 FC? 25

 66 GTO 39

 67 PSE

 68 STO M



STO ∙ M
 69 7

 70 OUTAN

 71 GTO 01

 72 RTN
 73_LBL "SENDM"


send matrix-file
 74 1.001

 75 IJ=A

 76 XEQ 20

 77 "D6543210"

 78_LBL 02

 79 SF 25

 80 C>+

 81 FC? 25

 82 GTO 39

 83 PSE

 84 STO M



STO ∙ M
 85 7

 86 OUTAN

 87 GTO 02

 88 RTN
 89_LBL "RECVX"


receive X-Register

 90 XEQ 20
 91 XEQ 30
 92 CLA
 93 7
 94 INAN



get data from HP84164A
 95 RCL M



RCL ∙ M
 96 GTO 41
 97 RTN
 98_LBL "RECVP"


receive program from EMU41
 99 XEQ 20
100 XEQ 30

101 INP

102 GTO 40
103 RTN

104_LBL "RECVR"


receive data register block
105 XEQ 20

106 XEQ 30

107_LBL 10

108 CLA

109 7

110 INAN

111 RDN

112 RCL M



RCL ∙ M
113 STO IND Y

114 RDN

115 ISG X

116 GTO 10

117 GTO 40

118 RTN

119_LBL "RECVF"


receive X-Memory data-file
120 CLX

121 SEEKPTA

122 XEQ 20

123 XEQ 30

124_LBL 11

125 CLA

126 7

127 INAN

128 RCL M



RCL ∙ M
129 SF 25

130 SAVEX

131 FS? 25

132 GTO 11

133 GTO 40

134 RTN
135_LBL "RECVM"


receive matrix-file
136 1.001

137 IJ=A

138 XEQ 20

139 XEQ 30

140_LBL 12

141 CLA

142 7

143 INAN

144 RCL M



RCL ∙ M
145 SF 25

146 >C+

147 FS? 25

148 GTO 12

149 GTO 40

150 RTN
151_LBL 20



initialisation routine for the HP82164A     )**
152 "HP82164A"

153 FINDID

154 SELECT

155 MANIO

156 CF 17

157 REMOTE

158 "LI5"



disable data terminal ready control signal line
159 OUTA



(R04-3)
160 "C0"



no software protocol (R11-3,2,1,0)
161 OUTA




162 NOTREM

163 RDN

164 RTN

165_LBL 30



waiting for data from HP82164A
166 ENTER^

167 CLX

168 X<>FIO

169_LBL 31

170 STAT

171 2

172 ATOXX

173 X<>FIO

174 RDN

175 FC? 02

176 GTO 31

177 X<>FIO

178 RDN

179 RTN
180_LBL 39



transfers dummy data
181 CLX

182 STO M
183 7

184 OUTAN

185_LBL 40



end routine clears Alpha and X-Register

186 CLX

187_LBL 41
188 AUTOIO

189 CLA

190 END
Software for EMU41
 01_LBL "COM"

 02_LBL "SENDX


send X-Register

 03 XEQ 20



initialisation SERIAL1
 04 "D6543210"


loaf dummy Bytes to Alpha
 05 STO M



STO ∙ M replaces dummy Bytes with 
 06 7




X-Register contents
 07 OUTAN



send Alpha-Register
 08 RDN

 09 GTO 41



end routine

 10 RTN
 11_LBL "SENDP"


send program
 12 XEQ 20



initialisation SERIAL1
 13 "PROG NAME ?"

asks for the program name
 14 AON

 15 PROMPT

 16 AOFF

 17 OUTP
 18 GTO 40

 19 RTN
 20_LBL "SENDR"


send data register block
 21 XEQ 20

 22 "D6543210"

 23_LBL 00

 24 RCL IND X

 25 XEQ 50

 26 STO M



STO ∙ M
 27 7

 28 OUTAN

 29 RDN

 30 RDN

 31 ISG X

 32 GTO 00

 33 GTO 40

 34 RTN

 35_LBL "SENDF"


send X-Memory data-file
 36 CLX

 37 SEEKPTA

 38 XEQ 20

 39 "D6543210"

 40_LBL 01

 41 SF 25

 42 GETX

 43 XEQ 50

 44 FC? 25

 45 GTO 39

 46 STO M



STO ∙ M
 47 7

 48 OUTAN

 49 GTO 01

 50 RTN
 51_LBL "SENDM"


send matrix-file
 52 1.001

 53 IJ=A

 54 XEQ 20

 55 "D6543210"

 56_LBL 02

 57 SF 25

 58 C>+

 59 FC? 25

 60 GTO 39

 61 XEQ 50

 62 STO M



STO ∙ M
 63 7

 64 OUTAN

 65 GTO 02

 66 RTN

 67_LBL "RECVX"


receive X-Register

 68 XEQ 20

 69 XEQ 30

 70 CLA

 71 7

 72.INAN



get data from COM
 73 RCL M



RCL ∙ M
 74 GTO 41

 75 RTN
 76_LBL "RECVP"


receive program
 78 16

 79 WSIZE

 80 13.1



calculates the absolute address of the
 81 PEEKB



first free Byte behind the last program
 82 15




in main memory
 83 AND

 84 256

 85 *

 86 13.0

 87 PEEKB

 88 X<>Y

 89 RDN

 90 +




absolute address
 91 ENTER^

 92 ENTER^

 93 0.2

 94 +

 95 PEEKB

 96 S>

 97 7

 98 AND



number of Bytes
 99 RCL Z

100 ENTER^

101 0.1

102 +

103 PEEKB



number of Register
104 7

105 *

106 RCL Z

107 +

108 X<>Y

109 INT

110 X<>Y

111 A+B

112 A-




absolute address of the first free Byte
113 XEQ 20

114 XEQ 30

115 ENTER^

116 CLA

117 7

118 INAN



read length of the program in Bytes
119 RDN

120 ENTER^

121 RCL M



RCL ∙ M
122 CHS

123 A+B

124 X<>Y

125_LBL 04

126 CLA

127 7

128 INAN



read program Byte from COM
129 RCL M



RCL ∙ M
130 X<>Y

131 RDN

132 POKEB



write program Byte to main memory
133 RDN

134 A-

135 X≠Y?

136 GTO 04

137 INT



add final END
138 1

139 -

140 "D____ _-"


"D 000 000 000 000 192 000 045"     )*
141 RCL M



RCL ∙ M
142 POKER

143 RDN



place new .END. position
144 13.1



to status register c
145 PEEKB

146 240

147 AND

148 256

149 *

150 RCL Z

151 +

152 "D"

153 XTOAH

154 13.0

155 ATOXR

156 POKEB

157 13.1

158 ATOXR

159 POKEB

160 GTO 40

161 RTN



after RECVP execute manual PACK = GTO ∙ ∙
162_LBL "RECVR"


receive data register block
163 XEQ 20

164 XEQ 30

165_LBL 10

166 CLA

167 7

168 INAN

169 RDN

170 RCL M



RCL ∙ M
171 STO IND Y

172 RDN

173 ISG X

174 GTO 10

175 GTO 40

176 RTN

177_LBL "RECVF"


receive X-Memory data-file
178 CLX

179 SEEKPTA

180 XEQ 20

181 XEQ 30

182_LBL 11

183 CLA

184 7

185 INAN

186 RCL M



RCL ∙ M
187 SF 25

188 SAVEX

189 FS? 25

190 GTO 11

191 GTO 40

192 RTN

193_LBL "RECVM"


receive matrix-file
194 1.001

195 IJ=A

196 XEQ 20

197 XEQ 30

198_LBL 12

199 CLA

200 7

201 INAN

202 RCL M



RCL ∙ M
203 SF 25

204 >C+

205 FS? 25

206 GTO 12

207 GTO 40

208 RTN

209_LBL 20



initialisation SERIAL1
210 "SERIAL1"

211 FINDID

212 SELECT

213 MANIO

214 CF 17

215 RDN

216 RTN

217_LBL 30



waiting for data from COM port
218 INSTAT



not any PC support this subroutine
219 RDN



than delete program lines 218 to 222
220 FC? 00

221 GTO 30

222 CF 00

223 RTN

224_LBL 39



transfer dummy data
225 CLX

226 STO M
227 7

228 OUTAN

229_LBL 40



end routine clears Alpha and X-Register
230 CLX

231_LBL 41
232 AUTOIO

233 CLA

234 RTN
235_LBL 50



time delay routine for slow speed PC
236 PSE



number of needed PSE commands depends on PC
237 END



and EMU41 speed mode
)* For   entering   the   synthetic   text   line   "D  000  000  000  000  192  000  045"  use    the

    CCD-Module :  Go to the  ALPHA  mode,  deactivate  USER , press  SHIFT  ENTER  and

     you  get  the  prompt  : _ _ _   for  the  input  of  the  decimal  values.

)** Set HP-IL/RS232 interface HP82164A to DTE mode (check the internal jumper position).

New EMU41 applications

Great thanks to Jean-Francois for expanding EMU41 with advanced RSU storage and RS232 gateway features. Jean-Francois has also added this new RS232 extension to EMU71. This is an encouraging result from a discussion during the last Allschwil meeting (HP-Collection) in Switzerland. In my own opinion EMU41 is the world´s leading HP41 emulator software because : 

EMU41 runs on different HP handheldcomputer- and PC platforms

EMU41 allows the rom-image formats .bin and .rom

EMU41 includes plug in modules and bank switched modules

EMU41 includes HEPAX, MBK-Profiset, RAM-BOX and RSU support

EMU41 includes a gateway for data exchange to other PC applications

EMU41 includes different gateway solutions for interfacing your real HP41

EMU41 includes some virtual IL devices like printer and mass storage
EMU41 includes virtual tapes for email exchange of data, programs, rom-images

EMU41 is able to control real HP-IL devices

May this development will result some more HP41 applications : For example an umbilical cord from EMU41, using RS232 + USB, to the MLDL2000 unit from Meindert Kuipers ?

Now two serial devices are available for EMU41





1, = COM port 1


E3 = default communication parameters : 9600 baud, no parity, 1 stop bit, 8 data bits








Christoph Klug,   Hildesheim/Germany,   Spring 2006

